
Exercise 1 (Theoretical Foundations): (6 + 4 = 10 points)

Let

ϕ = p(0) ∧ ∃X ¬p(X) ∧ ∀Y ( p(Y )→ p(s(s(Y ))) ) and

ψ = ∃Z ∀U ( p(U)→ p(s(Z)) )

be formulas over the signature (Σ,∆) with Σ = Σ0 ∪Σ1, Σ0 = {0}, Σ1 = {s}, and ∆ = ∆1 = {p}.

a) Prove that ϕ |= ψ by means of resolution.

Hint: First transform the formula ϕ∧¬ψ into a satisfiability-equivalent Skolem normal form, and
then convert this Skolem normal form to an equivalent clause set.

b) Explicitly give a Herbrand model of the formula ϕ (i.e., specify a carrier and a meaning for all
function and predicate symbols). You do not have to provide a proof for your answer.

Solution:

a)

ϕ ∧ ¬ψ
⇔ p(0) ∧ ∃X ¬p(X) ∧ ∀Y ( p(Y )→ p(s(s(Y ))) ) ∧ ¬∃Z ∀U ( p(U)→ p(s(Z)) )

⇔ p(0) ∧ ∃X ¬p(X) ∧ ∀Y (¬p(Y ) ∨ p(s(s(Y ))) ) ∧ ∀Z ∃U ¬(¬p(U) ∨ p(s(Z)) )

⇔ p(0) ∧ ∃X ¬p(X) ∧ ∀Y (¬p(Y ) ∨ p(s(s(Y ))) ) ∧ ∀Z ∃U ( p(U) ∧ ¬p(s(Z)) )

⇔ ∃X ∀Y ∀Z ∃U ( p(0) ∧ ¬p(X) ∧ (¬p(Y ) ∨ p(s(s(Y ))) ) ∧ p(U) ∧ ¬p(s(Z)) )

We obtain the following formula θ in Skolem normal form, which is satisfiability-equivalent to
ϕ ∧ ¬ψ (by dropping existential quantification, replacing X by c and replacing U by f(Y, Z)).

θ = ∀Y ∀Z ( p(0) ∧ ¬p(c) ∧ (¬p(Y ) ∨ p(s(s(Y ))) ) ∧ p(f(Y, Z)) ∧ ¬p(s(Z)) )

An equivalent clause set for the Skolem normal form θ of ϕ ∧ ¬ψ is:

{ {p(0)}, {¬p(c)}, {¬p(Y ), p(s(s(Y )))}, {p(f(Y, Z))}, {¬p(s(Z))} }

We perform resolution on this clause set to show unsatisfiability of θ and hence also of ϕ ∧ ¬ψ.

{p(0)} {¬p(c)} {¬p(Y ), p(s(s(Y )))} {p(f(Y, Z))} {¬p(s(Z))}

{¬p(Y )}
Z/s(Y )

�
Y/0

Hence, we have proven ϕ |= ψ. �

b) We have S |= ϕ for the Herbrand structure S = (T (Σ), α) with α0 = 0, αs(t) = s(t), and
αp = {s2i(0) | i ∈ N}.

.

Exercise 2 (SLD tree): (9 points)

Consider the following Prolog program P which can be used to divide two numbers represented by terms
built from 0 ∈ Σ0 and s ∈ Σ1 (i.e., 0 =̂ 0, 1 =̂ s(0), 2 =̂ s(s(0)), . . .):

div(X, 0, Z) :- !, fail.
div(0, s(Y), Z) :- eq(Z, 0), !.
div(X, Y, s(Z)) :- sub(X, Y, U), div(U, Y, Z).
sub(0, Y, 0).
sub(s(X), 0, s(X)).
sub(s(X), s(Y), Z) :- sub(X, Y, Z).
eq(X, X).

Please give a graphical representation of the SLD tree for the query ?- div(s(s(0)), s(0), X). in
the program P. Also give all answer substitutions explicitly.

Solution:

In this representation, the nodes and edges deleted by the cut are shown with a gray background and
dashed edges, respectively.

div(s(s(0)), s(0), X).

sub(s(s(0)), s(0), U), div(U, s(0), A).

X/s(A)

sub(s(0), 0, U), div(U, s(0), A).

div(s(0), s(0), A).

sub(s(0), s(0), U’), div(U’, s(0), B).

A/s(B)

sub(0, 0, U’), div(U’, s(0), B).

div(0, s(0), B).

. . .eq(B, 0), !.

!.

B/0

�

Answer substitution: X/s(s(0))

.

Exercise 3 (Fixpoint Semantics): (3 + 3 = 6 points)

Consider the following logic program P over the signature (Σ,∆) with Σ = {0, s} and ∆ = {plus}.

plus(X, 0, X).
plus(0, X, X).
plus(s(X), s(Y), s(s(Z))) :- plus(X, Y, Z).

a) For each n ∈ N explicitly give transnP(∅) in closed form, i.e., using a non-recursive definition.

b) Compute the set lfp(transP).

Solution:

Let G be the set of all ground terms, i.e., G = {si(0) | i ∈ N} = T (Σ).

a)

trans0P(∅) = ∅
trans1P(∅) = {plus(t, 0, t), plus(0, t, t) | t ∈ G}
trans2P(∅) = {plus(t, 0, t), plus(0, t, t), plus(t, s(0), s(t)), plus(s(0), t, s(t)) | t ∈ G}

...

transnP(∅) = {plus(t, si(0), si(t)), plus(si(0), t, si(t)) | t ∈ G, 0 ≤ i < n}
= {plus(si(t), si(0), s2i(t)), plus(si(0), si(t), s2i(t)) | t ∈ G, 0 ≤ i < n}

b) lfp(transP) = {plus(si(0), sj(0), si+j(0)) | i , j ∈ N}

.

Exercise 4 (Definite Logic Programming): (10 points)

A directed graph is a pair (V, E) where V is the set of vertices and E ⊆ V × V is the set of edges. We
represent such a graph by two lists where the first list contains the names of the nodes and the second
list contains an element e(n1,n2) iff there is an edge from n1 to n2 in the represented graph. Both lists
do not contain duplicates. For example, consider the following graph ({1, 2, 3}, {(1, 2), (3, 1), (2, 3)}):

1 2

3

This graph is represented by the two lists [1,2,3] and [e(1,2),e(3,1),e(2,3)].

Implement the predicate hamiltonian/2 in Prolog. A call hamiltonian(t1,t2) works as follows. If
both t1 and t2 are finite lists representing a directed graph as described above, the call succeeds iff there
is a hamiltonian cycle in the represented graph. If t1 and t2 are not of the form described above, then
hamiltonian(t1,t2) may behave arbitrarily. You must not use any built-in predicates in this exercise.
Note that meta-programming (e.g., using variables without a surrounding predicate) also uses built-in
predicates (implicitly) and is, thus, not allowed in this exercise.

A hamiltonian cycle exists in a directed graph iff there is a path through the graph starting and ending
in the same node (thus, visiting this node exactly twice) and visiting all other nodes exactly once. The
empty graph has no hamiltonian cycle. The following example calls to hamiltonian/2 illustrate its
definition:

• ?- hamiltonian([1,2,3],[e(1,2),e(3,1),e(2,3)]). succeeds with the empty answer sub-
stitution

• ?- hamiltonian([1,2,3],[e(1,2),e(2,1),e(1,3),e(3,1)]). fails

Hint: Use a helper predicate tour/4 with the following four arguments: (1) the node where you currently
are, (2) the list of nodes yet to visit, (3) the node you have to reach in the end, and (4) the list of
edges in the graph. Moreover, it can be helpful to define a predicate choice/3 for the non-deterministic
choice of the next node to visit. Here, choice(t1, t2, t3) is true if t1 is a list containing t2 and t3 is the
list t1 where the element t2 was deleted.

foo

Solution:

hamiltonian([N|NS],E) :- tour(N,NS,N,E).

tour(L,[],T,E) :- edge(L,T,E).
tour(L,[X|XS],T,E) :- choice([X|XS],Y,YS),

edge(L,Y,E),
tour(Y,YS,T,E).

choice([X|XS],X,XS).
choice([X|XS],Y,[X|YS]) :- choice(XS,Y,YS).

edge(L,T,[e(L,T)|_]).
edge(L,T,[_|E]) :- edge(L,T,E).

.

Exercise 5 (Arithmetic): (5 points)

Implement the predicate mean/2 in Prolog. A call of mean(t1,t2) works as follows. If t1 is a finite
non-empty list of integers, then t2 is unified with the rounded mean value of all numbers in t1. If t1 is
the empty list, mean/2 fails. If t1 is not a finite list of integers, mean/2 may behave arbitrarily.

We define the rounded mean value of n > 0 integers a1, . . . , an as
⌊

Σni=1ai
n

⌋
if

Σni=1ai
n

is positive and⌈
Σni=1ai
n

⌉
if

Σni=1ai
n

is negative.

The following example calls to mean/2 illustrate its definition:

• ?- mean([],X). fails

• ?- mean([1,2,3],1). fails

• ?- mean([1,2,3],X). succeeds with the answer substitution X = 2

• ?- mean([1,2],1). succeeds with the empty answer substitution

• ?- mean([1,-2],X). succeeds with the answer substitution X = 0

Hint: In Prolog, the term X // Y can be used to compute
⌊
X

Y

⌋
if
X

Y
is positive and

⌈
X

Y

⌉
if
X

Y
is

negative.

Solution:

mean(List,Mean) :- sumMean(0,0,List,Mean).

sumMean(Length,Sum,[],Mean) :- Length > 0,
Mean is Sum // Length.

sumMean(Length,Sum,[X|XS],Mean) :- NextLength is Length + 1,
NextSum is Sum + X,
sumMean(NextLength,NextSum,XS,Mean).

.

Exercise 6 (Meta-Programming): (10 points)

Implement the predicate fold/4 in Prolog. A call of fold(t1,t2,t3,t4) works as follows. If t1 is a
constant f ∈ Σ0 and t3 has the form [a1, . . . , an] with n > 0 (i.e. t3 is a finite non-empty list),
then the calls f (t2, a1, X1), f (X1, a2, X2), . . . , f (Xn−2, an−1, Xn−1), f (Xn−1, an, t4) with fresh Prolog
variables X1, . . . , Xn−1 are executed. That means we assume that there is also a predicate symbol
f ∈ ∆3 (with the same name as f ∈ Σ0). Thus, fold(f ,t2,[a1, . . . , an],t4) succeeds iff the query
f (t2, a1, X1), f (X1, a2, X2) . . . , f (Xn−2, an−1, Xn−1), f (Xn−1, an, t4) succeeds. If t1 or t3 are not of the
form described above, fold/4 may behave arbitrarily.
For example, the query ?- fold(foo,d,[a,b,c],X). is evaluated by executing the three calls
foo(d,a,X1), foo(X1,b,X2) and foo(X2,c,X). The query ?- fold(foo,b,[a],r). is evaluated by
the call foo(b,a,r).

Hint: You may use the built-in predicate =../2.

Solution:

fold(F,A,[E],Z) :- !,
C =.. [F,A,E,Z],
call(C).

fold(F,A,[E|ES],Z) :- C =.. [F,A,E,X],
call(C),
fold(F,X,ES,Z).

.

Exercise 7 (Constraint Logic Programming): (10 points)

A heterosquare is a matrix of dimension n × n containing all numbers from 1 to n2 such that the
respective sums of each row, of each column, and of each of the two diagonals are pairwise different.
For instance, consider the following matrix of dimension 3× 3: 1 2 4

3 7 8

5 6 9


Here the first row sums up to 7, the second row sums up to 18, the third row sums up to 20, the first
column sums up to 9, the second column sums up to 15, the third column sums up to 21, the top-left
to bottom-right diagonal sums up to 17, and the bottom-left to top-right diagonal sums up to 16. Since
all these eight values are pairwise different, this matrix indeed is a heterosquare.

We represent a heterosquare as a list of concatenated rows. For example, the above heterosquare would
be represented as follows:

[1, 2, 4, 3, 7, 8, 5, 6, 9]

Implement a Prolog predicate hsquare/1 such that the query ?- hsquare(L). has exactly those lists
L as answers that represent a heterosquare of dimension 3× 3. Thus, for a correct implementation we
get the following answers to the query (the order of the solutions depends on your implementation):

?- hsquare(L).
L = [1, 2, 3, 4, 5, 8, 6, 9, 7] ;
L = [1, 2, 3, 4, 5, 9, 6, 8, 7] ;
L = [1, 2, 3, 4, 5, 9, 7, 6, 8] ;

...

Hint: The query ?- hsquare(L). has more than 20,000 solutions.
Hint: You may use constraint logic programming for your implementation, but you are not required to
do so. Recall that the CLP library clpfd contains predicates like all_different/1, label/1, the infix
predicate ins/2, . . .

The following line is already given:

:- use_module(library(clpfd)).

10



Logic Programming WS 2010/2011
Solution - Exam 18.03.2011

Solution:

hsquare(Sq) :-
Sq = [E1, E2, E3, E4, E5, E6, E7, E8, E9],
Sq ins 1..9,
all_different(Sq),
Row1 #= E1 + E2 + E3,
Row2 #= E4 + E5 + E6,
Row3 #= E7 + E8 + E9,
Col1 #= E1 + E4 + E7,
Col2 #= E2 + E5 + E8,
Col3 #= E3 + E6 + E9,
Diag1 #= E1 + E5 + E9,
Diag2 #= E3 + E5 + E7,
all_different([Row1, Row2, Row3, Col1, Col2, Col3, Diag1, Diag2]),
label(Sq).

.

